
SAS syntax and output for Chapter 7: Basic Matrix Algebra for Statistical Modeling
Matrix algebra is more easily understood if one works through some calculations. The syntax in this document reproduces the matrix calculations presented in Chapter 7; cross-check the results produced by the calculations below against the calculations presented in Chapter 7.

Simple matrix operations
Define 3 x 2 matrix A with the ‘iml’ procedure; a comma denotes a new row:

	proc iml;
 reset print;
 A={1 3, 0.2 4.6, 8.5 -2.3};
quit;



‘reset print’ prints every result including matrix A to the results viewer:

	




Create A' and print the result to the console window:

	proc iml;
 reset print;
 A={1 3, 0.2 4.6, 8.5 -2.3};
 At = A`;
quit;





Now, take the transpose of At, i.e., (A')', and see that it is the same matrix as A:

	proc iml;
 reset print;
 A={1 3, 0.2 4.6, 8.5 -2.3};
 At = A`;
 Att = At`;
quit;





Define 3 x 3 matrix C:

	proc iml;
 reset print;
 C={-5 1 3, 2 2 6, 7 3 -4};
quit;





Calculate the trace of C:

	proc iml;
 reset print;
 C = {-5 1 3, 2 2 6, 7 3 -4};
 Ctr = trace(C);
quit;





See that C is not symmetric because C' is not the same as C:

	proc iml;
 reset print;
 C = {-5 1 3, 2 2 6, 7 3 -4};
 Ct = C`;
quit;





Define 3 x 3 matrix D:

	proc iml;
 reset print;
 D = {-5 1 3, 1 2 6, 3 6 -4};
quit;





D is symmetric because D' and D are the same:

	proc iml;
 reset print;
 D = {-5 1 3, 1 2 6, 3 6 -4};
 Dt = D`;
quit;





Matrix addition and subtraction
Define new 2 x 3 matrices A and B:

	proc iml;
 reset print;
 A = {1 2 3, 4 5 6};
 B = {-5 1 2, 3 0 -4};
quit;



Calculate C = A + B and check the result:

	proc iml;
 reset print;
 A = {1 2 3, 4 5 6};
 B = {-5 1 2, 3 0 -4};
 C = A + B;
quit;





Calculate D = A - B and check the result:

	proc iml;
 reset print;
 A = {1 2 3, 4 5 6};
 B = {-5 1 2, 3 0 -4};
 D = A - B;
quit;





Calculate E = -B and check the result:

	proc iml;
 reset print;
 A = {1 2 3, 4 5 6};
 B = {-5 1 2, 3 0 -4};
 E = -1 * B;
quit;





Multiplication with matrices
Define a new 3 x 2 matrix A:

	proc iml;
 reset print;
 A = {1 3, 2 5, 8 -2};
quit;



Calculate B = 3A and check the result:

	proc iml;
 reset print;
 A = {1 3, 2 5, 8 -2};
 B = 3 * A;
quit;





Define new 3 x 2 matrix A and new 2 x 2 matrix B:

	proc iml;
 reset print;
 A = {2 1, 4 2, 6 3};
 B = {1 3, 3 1};
quit;



Calculate matrix product C = AB:

	proc iml;
 reset print;
 A = {2 1, 4 2, 6 3};
 B = {1 3, 3 1};
 C = A * B;
quit;





What happens when you try to multiply A and B in the opposite order to form X = BA?

	proc iml;
 reset print;
 A = {2 1, 4 2, 6 3};
 B = {1 3, 3 1};
 X = B * A;
quit;



But you can calculate X = BA':

	proc iml;
 reset print;
 A = {2 1, 4 2, 6 3};
 B = {1 3, 3 1};
 X = B * A`;
quit;





Check the result; how does it compare to C = AB?

	proc iml;
 reset print;
 A = {2 1, 4 2, 6 3};
 B = {1 3, 3 1};
 X = B * A`;
 C = A * B;
quit;





Define new 2 x 2 matrices D and E:

	proc iml;
 reset print;
 D = {1 2, 3 4};
 E = {0 3, 2 1};
 quit;



Calculate matrix product F = DE and check the result:

	proc iml;
 reset print;
 D = {1 2, 3 4};
 E = {0 3, 2 1};
 F = D * E;
quit;





Notice that a different result is obtained if you pre-multiply D by E:

	proc iml;
 reset print;
 D = {1 2, 3 4};
 E = {0 3, 2 1};
 G = E * D;
quit;





Determinants and matrix inversion
Define a new 3 x 3 matrix A:

	proc iml;
 reset print;
 A = {3 1 4, 4 1 5, 2 3 5};
quit;



Get the determinant of A; confirm that A is singular:

	proc iml;
 reset print;
 A = {3 1 4, 4 1 5, 2 3 5};
 Ad = det(A);
quit;





Define a new 2 x 2 matrix A:

	proc iml;
 reset print;
 A = {2 9, 1 4};
quit;



Get the inverse of A using the 'inv' function and check the result:

	proc iml;
 reset print;
 A = {2 9, 1 4};
 Ainv = inv(A);
quit;





Confirm that Ainv is the inverse of A:

	proc iml;
 reset print;
 A = {2 9, 1 4};
 Ainv = inv(A);
 check1 = A * Ainv;
 check2 = Ainv * A;
quit;





Define 2 x 1 vector y and new 2 x 2 matrix A:

	proc iml;
 reset print;
 y = {-5, 7};
 A = {2 9, 1 4};
quit;



Solve for x as A-1y and check the result:

	proc iml;
 reset print;
 y = {-5, 7};
 A = {2 9, 1 4};
 x = inv(A) * y;
quit;





Matrix calculations for statistical applications
Means
Input data as a column vector X:

	proc iml;
 reset print;
 X = {3, 4, 6, 2, 5};
quit;



Define conformable unit vector unitV:

	proc iml;
 reset print;
 X = {3, 4, 6, 2, 5};
 N = nrow(X); /*the number of rows of X is the sample size*/
 unitV = j(1,N,1); /*create a row vector of ones with length N*/
quit;



Calculate the mean and check the result:

	proc iml;
 reset print;
 X = {3, 4, 6, 2, 5};
 N = nrow(X); 
 unitV = j(1,N,1); 
 meanX = (unitV * X)/N;
quit;





Input new data as an N x P (5 x 3) data matrix X:

	proc iml;
 reset print;
 X = {3 7.8 0, 
	  4 8.0 0,
	  6 8.2 1,
	  2 7.9 1,
	  5 8.1 1};
quit;



Calculate the vector of means and check the result:

	proc iml;
 reset print;
 X = {3 7.8 0, 
	  4 8.0 0,
	  6 8.2 1,
	  2 7.9 1,
	  5 8.1 1};
 N = nrow(X); 
 unitV = j(1,N,1); 
 meansX = (unitV * X)/N;
quit;





Variance and covariance
Calculate vector of deviation scores from X and check the result:

	proc iml;
 reset print;
 X = {3 7.8 0, 
	  4 8.0 0,
	  6 8.2 1,
	  2 7.9 1,
	  5 8.1 1};
 N = nrow(X); 
 unitV = j(1,N,1); 
 meansX = (unitV * X)/N;
 Xd = X - (unitV` * meansX);
quit;





Calculate the covariance matrix S and check the result:

	proc iml;
 reset print;
 X = {3 7.8 0, 
	  4 8.0 0,
	  6 8.2 1,
	  2 7.9 1,
	  5 8.1 1};
 N = nrow(X); 
 unitV = j(1,N,1); 
 meansX = (unitV * X)/N;
 Xd = X - (unitV` * meansX);
 S = (Xd` * Xd) / (N-1);
quit;





Standard deviation
Extract the variances from the covariance matrix. Rather than creating a diagonal matrix, the 'diag' operator will create a vector from the diagonal elements of D:

	proc iml;
 reset print;
 X = {3 7.8 0, 
	  4 8.0 0,
	  6 8.2 1,
	  2 7.9 1,
	  5 8.1 1};
 N = nrow(X); 
 unitV = j(1,N,1); 
 meansX = (unitV * X)/N;
 Xd = X - (unitV` * meansX);
 S = (Xd` * Xd) / (N-1);
 D = vecdiag(S);
quit;



Then calculate the standard deviations and check the result:

	proc iml;
 reset print;
 X = {3 7.8 0, 
	  4 8.0 0,
	  6 8.2 1,
	  2 7.9 1,
	  5 8.1 1};
 N = nrow(X); 
 unitV = j(1,N,1); 
 meansX = (unitV * X)/N;
 Xd = X - (unitV` * meansX);
 S = (Xd` * Xd) / (N-1);
 D = vecdiag(S);
 stdevs = D##.5;
quit;





Correlations
First, need to place the standard deviations obtained above back into a diagonal matrix, then get its inverse to form D-1/2:

	proc iml;
 reset print;
 X = {3 7.8 0, 
	  4 8.0 0,
	  6 8.2 1,
	  2 7.9 1,
	  5 8.1 1};
 N = nrow(X); 
 unitV = j(1,N,1); 
 meansX = (unitV * X)/N;
 Xd = X - (unitV` * meansX);
 S = (Xd` * Xd) / (N-1);
 D = vecdiag(S);
 stdevs = D##.5;

 D = j(nrow(stdevs),nrow(stdevs),0); 
 start SetDiag(A, v); /* set diagonal elements */
   diagIdx = do(1,nrow(A)*ncol(A), ncol(A)+1);
   A[diagIdx] = v;             
 finish;
 run SetDiag(D, stdevs); 

 Dinv = inv(D);
quit;



Then calculate the correlation matrix and check the result:

	proc iml;
 reset print;
 X = {3 7.8 0, 
	  4 8.0 0,
	  6 8.2 1,
	  2 7.9 1,
	  5 8.1 1};
 N = nrow(X); 
 unitV = j(1,N,1); 
 meansX = (unitV * X)/N;
 Xd = X - (unitV` * meansX);
 S = (Xd` * Xd) / (N-1);
 D = vecdiag(S);
 stdevs = D##.5;

 D = j(nrow(stdevs),nrow(stdevs),0); 
 start SetDiag(A, v); /* set diagonal elements */
   diagIdx = do(1,nrow(A)*ncol(A), ncol(A)+1);
   A[diagIdx] = v;             
 finish;
 run SetDiag(D, stdevs); 

 Dinv = inv(D);
 R = Dinv * S * Dinv;
quit;





Eigenvalues and eigenvectors
We can see the eigenstructure of the correlation matrix calculated above using the 'call eigen' function:

	proc iml;
 reset print;
 X = {3 7.8 0, 
	  4 8.0 0,
	  6 8.2 1,
	  2 7.9 1,
	  5 8.1 1};
 N = nrow(X); 
 unitV = j(1,N,1); 
 meansX = (unitV * X)/N;
 Xd = X - (unitV` * meansX);
 S = (Xd` * Xd) / (N-1);
 D = vecdiag(S);
 stdevs = D##.5;

 D = j(nrow(stdevs),nrow(stdevs),0); 
 start SetDiag(A, v); /* set diagonal elements */
   diagIdx = do(1,nrow(A)*ncol(A), ncol(A)+1);
   A[diagIdx] = v;             
 finish;
 run SetDiag(D, stdevs); 
 Dinv = inv(D);
 R = Dinv * S * Dinv;

 call eigen(val, rvec, R);
quit;





The following code confirms that R = UDU' where U is a matrix formed by the three eigenvectors and D is a diagonal matrix of the eigenvalues:

	proc iml;
 reset print;
 X = {3 7.8 0, 
	  4 8.0 0,
	  6 8.2 1,
	  2 7.9 1,
	  5 8.1 1};
 N = nrow(X); 
 unitV = j(1,N,1); 
 meansX = (unitV * X)/N;
 Xd = X - (unitV` * meansX);
 S = (Xd` * Xd) / (N-1);
 D = vecdiag(S);
 stdevs = D##.5;

 D = j(nrow(stdevs),nrow(stdevs),0); 
 start SetDiag(A, v); /* set diagonal elements */
   diagIdx = do(1,nrow(A)*ncol(A), ncol(A)+1);
   A[diagIdx] = v;             
 finish;
 run SetDiag(D, stdevs); 
 Dinv = inv(D);
 R = Dinv * S * Dinv;

 call eigen(val, rvec, R) ;
 
 D = j(nrow(val),nrow(val),0); 
 start SetDiag(A, v); /* set diagonal elements */
   diagIdx = do(1,nrow(A)*ncol(A), ncol(A)+1);
   A[diagIdx] = v;             
 finish;
 run SetDiag(D, val); 
 R = rvec * D * rvec`;
quit;





The matrix R above is the same as the correlation matrix calculated earlier.

Matrix equations for linear regression
Below, we use the data file 'generic.csv' to confirm that the matrix formulas presented in Chapter 7 for OLS multiple regression produce the same results as the regular 'reg' procedure used to estimate a regression model in SAS.

	data dat;
 infile 'c:\generic.csv' FIRSTOBS=2 dlm=',';
 input const x1 x2 x3 y;
run;



Take a look at the first 6 rows of the data:

	proc print data=dat (obs=6);
run;





In this example, we would like to use the OLS matrix formula to estimate the parameters (intercept and slope coefficients) for the linear regression of 'y' on 'x1', 'x2', and 'x3'.
Notice above that a column of 1s (called 'const' for 'constant') is included for your convenience: As explained in Chapter 7, it is important for the constant to be inluded in X so that the intercept parameter can be estimated.
First, need to partition the data set to obtain a separate data matrix X for the predictors (including the constant) and Y for the outcome variable:

	proc iml;
 use dat;
 read all;
 Y = y;
 X = const || x1 || x2 || x3;
quit;



Calculate the regression parameter estimates and check the result:
	proc iml;
 reset print;
 use dat;
 read all;
 Y = y;
 X = const || x1 || x2 || x3;
 Xt = X`;
 betaHat = inv(Xt * X) * Xt * Y;
quit;




 
The output above gives the estimated intercept as 3.42, estimated coefficient for x1 is 0.23, coefficient for x2 is 0.47, coefficient for x3 is 0.52.
Calculate the vector of predicted values and check the result:

	proc iml;
 reset print;
 use dat;
 read all;
 Y = y;
 X = const || x1 || x2 || x3;
 Xt = X`;
 betaHat = inv(Xt * X) * Xt * Y;
 Yhat = X * betaHat;
quit;





Calculate the vector of predicted values using the hat matrix, H and check that the result is the same as above:

	proc iml;
 reset print;
 use dat;
 read all;
 Y = y;
 X = const || x1 || x2 || x3;
 Xt = X`;
 betaHat = inv(Xt * X) * Xt * Y;
 Yhat = X * betaHat;
 H = X * inv(Xt * X) * Xt;
 Yhat = H * Y;
quit;







Extract leverage values from the hat matrix:

	proc iml;
 reset print;
 use dat;
 read all;
 Y = y;
 X = const || x1 || x2 || x3;
 Xt = X`;
 betaHat = inv(Xt * X) * Xt * Y;
 Yhat = X * betaHat;
 H = X * inv(Xt * X) * Xt;
 Yhat = H * Y;
 HatValues = vecdiag(H);
quit;







Calculate the residuals and check the result; confirm that the mean of the residuals is 0 (within a very small amount of rounding error):

	proc iml;
 reset print;
 use dat;
 read all;
 Y = y;
 X = const || x1 || x2 || x3;
 Xt = X`;
 betaHat = inv(Xt * X) * Xt * Y;
 Yhat = X * betaHat;
 H = X * inv(Xt * X) * Xt;
 Yhat = H * Y;
 HatValues = vecdiag(H);
 e = Y - Yhat;
 meane = round(e[:]);
quit;










Next, return to the original dataset to use the 'reg' procedure to estimate the regression model:

	proc reg data=dat plots=none;
 model y = x1 x2 x3 / clb;
 output out=temp r=resid h=hatval rstudent=rstudent cookd=cookd p=predict;
run;
quit;



Confirm that the parameter estimates, predicted values, leverage values, and residuals match the results calculated above using matrix formulas:

	


proc print data = temp;
 var predict;
run;



proc print data = temp;
 var hatval;
run;



proc print data = temp;
 var resid;
run;

